


Butterflies, Heat Maps, and More: Using the SAS/GRAPH® Procedures SGPLOT and SGPANEL


Susan Schwartz, SAS Institute Inc.
SAS Global Forum 2008

BUTTERFLY PLOTS




```
proc sgplot data=cancer;
  hbar cause / response=mcases fillattrs=graphdata1
    transparency=.65
    legendlabel="New Cases (Male)"
    name="mcases";
  hbar cause / response=mdeaths fillattrs=graphdata1
    transparency=.25 barwidth=.5
    legendlabel="Deaths (Male)"
    name="mdeaths";
  hbar cause / response=fcases fillattrs=graphdata2
    transparency=.65
    legendlabel="New Cases (Female)"
    name="fcases";
  hbar cause / response=fdeaths fillattrs=graphdata2
    transparency=.25 barwidth=.5
    legendlabel="Deaths (Female)"
    name="fdeaths";
  keylegend "mcases" "fcases" "mdeaths" "fdeaths"
    / across=2;
  xaxis label="" grid;
  yaxis label="" discreteorder=data;
run;
```

```
proc sgpanel data=cancer;
  panelby year / columns=1;
  hbar cause / response=mcases fillattrs=graphdata1
    transparency=.65
    legendlabel="New Cases (Male)"
    name="mcases";
  hbar cause / response=mdeaths fillattrs=graphdata1
    transparency=.25 barwidth=.5
    legendlabel="Deaths (Male)"
    name="mdeaths";
  hbar cause / response=fcases fillattrs=graphdata2
    transparency=.65
    legendlabel="New Cases (Female)"
    name="fcases";
  hbar cause / response=fdeaths fillattrs=graphdata2
    transparency=.25 barwidth=.5
    legendlabel="Deaths (Female)"
    name="fdeaths";
  keylegend "mcases" "fcases" "mdeaths" "fdeaths" /
    across=2;
  colaxis label="" grid;
  rowaxis label="" discreteorder=data;
run;
```



RISK MAPS (STACKED BAND PLOTS)


Body Mass Index Plot †



```
proc sgplot data=merged;
  band x=weight_bmi
    upper=upper_hgt lower=lower_hgt /
    transparency=.5
  group=category
  name="bmi";
  xaxis min=30 max=300 label='Weight (lbs)' grid;
  yaxis min=40 max=80 label='Height (in)' grid;
  keylegend "bmi" /
  position=right
  across=1
  title='BMI';
  series x=wtg y=hgt /
  datalabel=year
  lineattrs=(pattern=solid thickness=2px)
  markerattrs=(symbol=circlefilled) markers
  group=name
  name="pts";
  keylegend "pts" / position=bottom title='Patient: ';
  inset "BMI=703 x Weight / Height (*ESC*){sup '2'}" "" /
  position=bottomright
  ltextattrs=graphfootnotetext;
run;
```

Hypertension Plot


```
proc sgplot data=bp;
  band x=dias lower=losys upper=hisys /
  group=grp name="bp";
  series x=refx y=refy / lineattrs=(thickness=2);
  xaxis values=(75 to 115 by 5)
  label='Diastolic Pressure (mmHg)';
  yaxis values=(110 to 190 by 10)
  label='Systolic Pressure (mmHg)';
  keylegend "bp" / position=right across=1 title='Stage';
  inset "Isolated" "Systolic" "Hypertension" /
  position=left textattrs=graphlabeltext;
run;
```



† $\text{Body Mass Index (BMI)} = 703 \times \frac{\text{weight (lb)}}{\text{height}^2 (\text{in}^2)}$

Underweight = $\text{BMI} < 18.5$
 Normal = $18.5 \leq \text{BMI} < 25.0$
 Overweight = $25.0 \leq \text{BMI} < 30.0$
 Obese = $\text{BMI} \geq 30.0$


HEAT MAP




```
proc sgplot data=heatmap;
  scatter x=make y=type /
    group=group
    markerattrs=(size=.28in symbol=squarefilled);
run;
```

NEEDLE PLOT


```
proc sgplot data=yield cycleattrs;
  series x=mat1yr y=yld1yr / markers
  markerattrs=
    (symbol=circlefilled size=10px)
  lineattrs=(pattern=solid) name='n1';
  series x=mat1mo y=yld1mo / markers
  markerattrs=
    (symbol=circlefilled size=10px)
  lineattrs=(pattern=solid) name='n2';
  series x=matcur y=yldcur / markers
  markerattrs=
    (symbol=circlefilled size=10px)
  lineattrs=(pattern=solid) name='n3';
  needle x=matcur y=yldcur /
  lineattrs=graphreference;
  xaxis type=linear values=(1 2 3 6 7 8 9 11 14)
  label='Bond Maturity';
  yaxis grid label='Bond Yield';
  keylegend 'n1' 'n2' 'n3' /
  position=top noborder;
run;
```



PANELED PLOTS



```
proc sgpanel data=cancer;
  panelby race sex /
    columns=5
    layout=lattice
    novaname
    uniscale=column;
  hbar cause / response=incidence;
  colaxis grid label='Incidence per 100,000 People';
  rowaxis grid label='';
run;
```



```
proc sgpanel data=cancer;
  panelby cause sex /
    columns=5
    layout=lattice
    novaname
    uniscale=column;
  hbar race / response=incidence;
  colaxis grid label='Incidence per 100,000 People';
  rowaxis grid label='';
run;
```

```
proc sgpanel data=sashelp.heart;
  format cholesterol cholfmt. diastolic diasfmt.;
  panelby cholesterol sex /
    columns=3 rows=2
    novaname;
  vbar diastolic /
    response=systolic
    stat=mean;
  refline 140 /
    axis=y name="refline"
    lineattrs=(color=black)
    legendlabel=
    'Systolic Hypertension (140 mmHg)';
  colaxis grid label='Diastolic BP (mmHg)';
  rowaxis grid label='Systolic BP (mmHg)';
run;
```

